TROI SERIAL PLUG-IN™ 1.0 USER GUIDE

February 1999

Troi Automatisering
Vuurlaan 18
2408 NB Alphen a/d Rijn
The Netherlands
Tel: +31-172-426606
Fax: +31-172-470539

You can also visit the Troi web site at: <http://www.troi.com/> for additional information.

Table of Contents

Installing plug-ins

For Macintosh:

- Quit FileMaker Pro.
- Put the file "Troi Serial Plug-in" from the folder "MacOS" into the "FileMaker Extensions" folder in the FileMaker Pro 4 folder.
- If you have installed previous versions of this plug-in, you are asked: “An older item named “Troi Serial Plug-In” already exists in this location. Do you want to replace it with the one you’re moving?’. Press the OK button.
- Start FileMaker Pro. The first time the Troi Serial Plug-in is used it will display a dialog box, indicating that it is loading and showing the registration status.



For Windows:

- Quit FileMaker Pro.
- Put the file "trserial.fmx" from the directory "Windows" into the "SYSTEM" subdirectory in the FileMaker Pro 4.0 directory.
- If you have installed previous versions of this plug-in, you are asked: “This folder already contains a file called 'trserial.fmx'. Would you like to replace the existing file with this one?’. Press the Yes button.
- Start FileMaker Pro. The Troi Serial Plug-in will display a dialog box, indicating that it is loading and showing the registration status.



TIP You can check which plug-ins you have loaded by going to the plug-in preferences: Choose **Preferences** from the **Edit** menu, and then choose **Plug-ins**.

You can now open the file "SeriExpl.fp3" to see how to use the plug-in's functions. There is also a Function overview in this file.

IMPORTANT There is a problem in FileMaker Pro 4.0v1. Please make sure that all plug-ins that are in the folder "FileMaker Extensions" are enabled in the preferences. (Under Edit/ Preferences/ Application/ Plug-ins). Make sure all plug-ins have a cross before their name. Remove plug-ins you don't use from the "FileMaker Extensions" folder.
NB: This bug is fixed in version 4.1 and 4.0v2. So please upgrade to these versions.

If You Have Problems

This user manual tries to give you all the information necessary to use this plug-in. So if you have a problem please read this user guide first. If that doesn't help you can get free support by email. Send your questions to support@troi.com with a full explanation of the problem. Also give as much relevant information (version of the plug-in, which platform, version of the operating system, version of FileMaker Pro) as possible.

If you find any mistake in this manual or have a suggestion please let us know. We appreciate your feedback!

Summary of functions

Plug-ins add new functions to the standard functions that are available in FileMaker Pro. You can see those extra functions for all plug-ins at the top right of the Specify Calculation Box:



IMPORTANT In the United States, commas act as list separators in functions. In other countries semi-colons might be used as list separators. The separator being used depends on the operating system your computer uses, as well as the one used when the file was created. All examples show the functions with commas.

The Troi Serial Plug-in adds the following functions:

function name	short description
Serial-Version	check for correct version of the plug-in
Serial-GetPortsNames	returns the names of all serial ports that are available on the computer
Serial-Open	opens a serial port
Serial-Close	closes a serial port
Serial-Receive	receives data from a serial port
Serial-Send	send data to a serial port
Serial-SetDispatchScript	tell the plug-in which script to call when data is received
Serial-DataWasReceived	returns if data was received on a open port
Serial-RestoreSituation	tell the plug-in to bring the original file back to the front
Serial-ToASCII	converts (one or more) numbers to their equivalent ASCII characters

Using external functions

External functions for this plug-in can be used in a script step using a calculation. The external functions should not be used in a define field calculation.

IMPORTANT The Balance functions have to be used in a specific way, to create the desired effect. See the section on Balance functions for the specifics on this.

Serial-Version

Example usage: External(Serial-Version; "") will return "Troi Serial Plug-in 1.0b1".

IMPORTANT You should always check if the plug-in is loaded, by using this function. If the plug-in is not loaded use of external functions may result in unexpected result or data loss, as FileMaker will return an empty field to any external function that is not loaded.

Serial-GetPortNames

Syntax External("Serial-GetPortNames" , "")

Returns the names of all serial ports that are available on the computer.

Parameters

no parameters leave empty for future use.

Result

The returned result is a list of serial ports that are available on the computer that is running FileMaker Pro. Each available port is on a different line. On a desktop Mac a typical result will be:

Printer Port¶
Modem Port¶

On a portable Mac a typical result will be:

Printer-Modem Port¶
Internal Modem¶

On Windows the result will be:

COM1¶
COM2¶
COM3¶
COM4¶

Use this function to let the user of the database choose which port to open. Store the name of the chosen port in a global field. You can then check the next time the database is opened whether the portname is still present and ask the user if he wants to change his preference.

If an error occurs an error code is returned. Returned error codes can be:

$$-108 memFullErr Ran out of memory

Other errors might be returned.

NOTE On Windows currently there is no apparent way to test for the available portnames, so at the moment this function always returns the same result.

Serial-Open

Syntax Set Field[gErrorCode, External("Serial-Open" , "*portname* | *switches* ")]

Opens a serial port with this name and the specified parameters.

Parameters

portname: the name of the port to open
switches: (optional) specifies the setting of the port like the speed of the port etc.

Result

Returned result is an error code:

0	no error	
$$-50	paramErr	There was an error with the parameter
$$-108	memFullErr	Ran out of memory
$$-97	portInUse	Could not open port, the port is in use
$$-4210	portDoesnotExistErr	A serial port with this name is not available on this computer
$$-4211	AllPortsNullErr	No serial ports are available on this computer

Other errors might be returned.

Example usage

```
Set Field[ gErrorCode, External("Serial-Open" , "COM2 | baud=19200") ]
```

will open the COM2 port with a speed of 19200 baud.

Specifying the port settings

Default port settings

A serial port can be configured in a lot of ways. These settings can be set by specifying switches. If you don't specify any switches the port is initialised to the following settings: a speed of 9600 baud, no parity, 8 data bits, 1 stop bit, no handshaking. If you want to use this setting open the port like this:

```
Set Field[gErrorCode, External("Serial-Open", "COM2") ]
```

Specifying other port settings

It is recommended that you set the port settings explicitly.. Give the settings by concatenating the desired settings keywords. You specify them like this:

```
Set Field[gErrorCode, External("Serial-Open",
                    "COM2 | baud=9600 parity=none data=8 stop=10 flowControl=XOnXOff") ]
```

You can set the speed, the parity, the number of data and stopbits, and the handshaking to use. Note that the order of the keywords and case are ignored. All keywords are optional and should be separated by a space or a return.

Specifying the port speed

The port speed indicates how quick a the data is transported over the serial line.
Allowed values for the port speed are:

```
baud=150    baud=1800    baud=7200    baud=28800    baud=115200
baud=300    baud=2400    baud=9600    baud=38400    baud=230400
baud=600    baud=3600    baud=14400   baud=57600
baud=1200   baud=4800    baud=19200
```

NOTE Not all speeds may be supported on all serial ports. Check the documentation of the computer and the equipment you want to connect.

You need to specify the same speed that the other equipment is using. Higher port speeds can result in loss of data if the serial cable can't cope with this speed. If this happens try a lower speed.

Specifying the bit format options

Data over a serial port is send in small packet of 4 to 10 bits. This packet consists of 4-8 data bits, followed by a parity bit and stopbits.

Data bits
You can specify the number of the data bits by adding one of the datasize keywords to the switch parameter. The most used value is 8 data bits. Allowed values for the number of data bits are:

```
data=4      data=7
data=5      data=8
data=6
```

Parity bits
You can specify the parity bit by giving adding one of the following keywords to the switch parameter:

```
parity=none   parity=odd   parity=even
```

Stop bits
You can specify the number of stopbits by giving adding one of the following keywords to the switch parameter:

```
stop=10      stop=15      stop=20
```

Here `stop=10` means 1 stop bit, `stop=15` means 1.5 stopbit and `stop=20` means 2 stopbits.

Specifying the handshaking options

Handshaking is a way to ensure that the transfer of data can be stopped temporarily. This also called (data) flow control. A serial port can use hardware handshaking and software handshaking. For hardware handshaking to work the serial cable must have wires to support it.

Using the Serial-Open function this plug-in allows a basic way to set the handshaking and also an advanced way, which gives more options, but most users probably don't need.

Basic handshaking options

Basic handshaking has 3 keywords:

```
flowControl=DTRDSR      flowControl=RTSCTS      flowControl=XOnXOff
```

You can specify one or more of these flow control keywords. You should specify at least one of these keywords. Try `flowControl=DTRDSR` as this is mostly supported. `FlowControl=DTRDSR` and `flowControl=RTSCTS` are hardware handshaking options, for which you need proper cabling. `FlowControl=XOnXOff` is a software based handshake option.

`FlowControl=DTRDSR` means that the signal DTR is used for input flow control and DSR for output flow control. `FlowControl=RTSCTS` means that the signal RTS is used for input flow control and CTS for output flow control. `FlowControl=XOnXOff` uses a XOff character (control-S) and a XOn character (control-Q) to stop input and output flow.

IMPORTANT Do not use `FlowControl=XOnXOff` if you want to transfer binary data, like pictures. This protocol uses two ASCII characters that might also be in the binary data. `FlowControl=XOnXOff` works fine with normal text.

Example 1

```
Set Field[gErrorCode, External("Serial-Open",
                      "COM2 | baud=9600 parity=none data=8 stop=10 flowControl=DTRDSR") ]
```

This will set the port to use `DTR/DSR` hardware handshaking.

Example 2

```
Set Field[gErrorCode,   External("Serial-Open",
                        "COM2 | baud=9600 parity=none data=8 stop=10 flowControl=DTRDSR
                        flowControl=RTSCTS flowControl=XOnXOff") ]
```

This will set the port to use all 3 types of handshaking in parallel.

Advanced handshaking options

Advanced handshaking options allows you more control over the serial port settings. It enables you to set the handshaking of the output an input separately.

With advanced handshaking you can use the following keywords:

keyword	meaning
inputControl=XOnXOff	use XOnXOff for input handshaking
outputControl=XOnXOff	use XOnXOff for output handshaking
inputControl=RTS	use RTS for input handshaking
outputControl=CTS	use CTS for output handshaking
inputControl=DTR	use DTR for input handshaking
outputControl=DSR	use DSRfor output handshaking
DTR=enabled	set DTR signal permanent to high
DTR=disabled	set DTR signal permanent to low
RTS=enabled	set RTS signal permanent to high
RTS=disabled	set RTS signal permanent to low

Below you find how the basic handshaking keywords relate to the advanced handshaking keywords:

basic keyword	=	the same as 2 advanced keywords
flowControl=XOnXOff	=	inputControl=XOnXOff outputControl=XOnXOff
flowControl=RTSCTS	=	inputControl=RTS outputControl=CTS
flowControl=DTRDSR	=	inputControl=DTR outputControl=DSR

The other advanced keywords don't have a equivalent.

NOTE You can mix the basic handshaking keywords with the advanced handshaking keywords, as long as this is sensible.

Example 1

If you want to use DTR handshaking for input flow control and CTS for output flow control use the following settings to open COM1:

```
Set Field[gErrorCode,   External("Serial-Open",
                        "COM1 | baud=9600 parity=none data=8 stop=10
                         outputControl=CTS inputControl=DTR") ]
```

Example 2

If you want to enable the DTR signal and use XOnXOff input flow control use the following settings to open COM1:

```
Set Field[gErrorCode,   External("Serial-Open",
                        "COM1 | baud=9600 parity=none data=8 stop=10
                         DTR=enabled inputControl=XOnXOff") ]
```

Example 3

```
Set Field[gErrorCode, External("Serial-Open",
                        "COM2 | baud=9600 data=7 parity=odd stop=20 flowControl=XOnXOff
                         outputControl=CTS inputControl=DTR") ]
```

This shows that XOnXOff is used for input and output flow control and also DTR handshaking for input flow control and CTS for output flow control.

Serial-Close

Syntax Set Field[gErrorCode, External("Serial-Close" , "*portname*")]

Closes a serial port with the specified name . If the portname parameter is "" ALL ports are closed.

Parameters

portname: the name of the port to close

Result

The returned result is an error code:

0	no error	the port was closed
$$-4210	portDoesnotExistErr	A serial port with this name is not available on this computer
$$-4211	AllPortsNullErr	No serial ports are available on this computer
$$-108	memFullErr	Ran out of memory

Other errors might be returned.

Example Usage

This will close the COM3 port:

```
Set Field[ gErrorCode, External("Serial-Close" , "COM3") ]
```

This will close all open ports:

```
Set Field[ gErrorCode, External("Serial-Close" , "") ]
```

Serial-Receive

Syntax Set Field[gResult, External("Serial-Receive" , "*portname*")]

Receives data from a serial port with the specified name . The port needs to be opened first (See Serial-Open). If no data is available an empty string is returned:"".

Parameters

portname: the name of the port to receive data from

Result

The returned result is the data received or an error code. An error always starts with 2 dollars, followed by the error code. You should always check for errors when receiving by testing if the first two characters are dollars. See below.

Returned error codes can be:

$$-28	notOpenErr	The port is not open
$$-108	memFullErr	Ran out of memory
$$-50	paramErr	There was an error with the parameter
$$-4210	portDoesnotExistErr	A serial port with this name is not available on this computer
$$-4211	AllPortsNullErr	No serial ports are available on this computer
$$-207	notEnoughBufferSpace	The input buffer is full

Other errors might be returned.

Example Usage

```
Set Field[ gResult, External("Serial-Receive" , "Modem port") ]
```

This will receive data from the Modem port.

Example: Receiving and Testing for Errors

Below you find a "`Receive Data`"script for receiving data into a global text field `gTempResultReceived` , The script tests for errors. **`gPortName`** is a global text field where the name of the previously opened port was stored.

```
Set Field [gTempResultReceived, External("Serial-Receive", gPortName) ]
If [Left(gTempResultReceived, 2 ) = "$$"]
     Beep
     If [gTempResultReceived = "$$-28"]
          Show Message [Open the port first]
     Else
          If [gTempResultReceived = "$$-207"]
               Show Message [Buffer overflow error.]
          Else
               Show Message [An error occurred!]
          End If
     End If
     Halt Script
End If
```

Serial-Send

Syntax Set Field[gResult, External("Serial-Send" , "*portname* | *data*")]

Sends data to the serial port with the specified name . The port needs to be opened first (See Serial-Open).

Parameters

portname: the name of the port to send data to
data: the text data that is to be sent to the serial port

Result

The returned result is an error code. An error always starts with 2 dollars, followed by the error code. You should always check for errors when sending by testing if the first two characters are dollars. See below.

Returned error codes can be:

0	no error	the data was send
$$-28	notOpenErr	The port is not open
$$-108	memFullErr	Ran out of memory
$$-50	paramErr	There was an error with the parameter
$$-4210	portDoesnotExistErr	A serial port with this name is not available on this computer
$$-4211	AllPortsNullErr	No serial ports are available on this computer
$$-207	notEnoughBufferSpace	The output buffer is full

Other errors might be returned.

Example Usage

```
Set Field[ gResult, External("Serial-Send" ,
                  "Modem port| So long and thanks for all the fish") ]
```

This will send the string `"So long and thanks for all the fish"` to the Modem port.

Example: Sending and Testing for Errors

Below you find a "`Send Data`" script for sending data from a global text field **`gTempResultReceived`**, The script tests for errors. **`gPortName`** is a global text field where the name of the previously opened port was stored.

```
Set Field [gErrorCode, External("Serial-Send",  gPortName & "|" & gTextToSend) ]
If [Left(gErrorCode, 2 ) = "$$"]
     Beep
     If [gErrorCode = "$$-28"]
          Show Message [Open the port first]
     Else
          If [gErrorCode = "$$-207"]
               Show Message [Buffer overflow error.]
          Else
               Show Message [An error occurred while sending!]
          End If
     End If
     Halt Script
End If
```

Receiving data via Dispatch Scripting™

The dilemma: FileMaker currently doesn't have a easy, cross platform way to handle an event, like when data is received. To make it nevertheless possible we have come up with what we call Dispatch Scripting™.

This plug-in has a simple and cross platform way to execute a script when data has been received. This is done via a Dispatch Script. If you want this functionality you need to implement the Dispatch functions in your database. This is how this can be done:

During development

You have to implement this once:

- write the Dispatch Script or change an existing script
- include the Dispatch Script in the menu, so it can be called from the keyboard with control-1 to control 9 (Windows) or command-1 to command-9 (Mac)
- write a "Start receiving script" that
    - opens the serial port
    - and tells the plug-in which is the Dispatch Script.

When Running the database

When the database is running and you want to begin receiving:

- perform the "Start receiving script".

This tells the plug-in for example that the Dispatch Script can be called from the keyboard with control-1 (Windows) or command-1 (Mac).

This is what happens when data arrives:

- the plug-in will bring the database file to the front and simulate a press on the keyboard:control-1 (Windows) or command-1(Mac).
- this will start the Dispatch Script, which can handle the receiving of the data.

NOTE You can still use the Dispatch Script for other actions, so this doesn't cost a place in the menu. That's why we call it a dispatching script: when called it determines if it was called because there was data received and if yes it will dispatch the processing.

Functions to implement Dispatch Scripting

The following external functions help in achieving the receiving of data via the Dispatch Script.

Serial-SetDispatchScript	tell the plug-in which (Dispatch) script to call when data is received
Serial-DataWasReceived	returns 1 when data was received on a open port
Serial-RestoreSituation	tell the plug-in to bring the original file back to the front

See the sample file **Dispatch.fp3** for a working example.

Example Dispatch Script

Below you find a sample "To Menu" Dispatch Script:

```
If [External("Serial-DataWasReceived", "")]
     Perform Script [Sub-scripts, "Process Data Received"]
Else
     Enter Browse Mode []
     Go to Layout [“Menu”]
     Halt Script
End If
```

This script checks if there is data received. If this is the case it dispatches to the script "Process Data Received" which receives the data and puts it into a field.Else it will do its normal business (going to a menu).

Make sure you include this script in the menu. We assume this script can be performed with the keyboard shortcut :control-1 (Windows) or command-1 (Mac)

Example Process Data Received Script

Below you find a sample "Process Data Received" script, which gets the data from the plug-in into the field **mesReceived**.

```
Enter Browse Mode []
Perform Script [Sub-scripts, "Receive Data in global gTempResultReceived"]
Set Field [mesReceived, mesReceived & gTempResultReceived]
Set Field [gErrorCode, External("Serial-RestoreSituation", "") ]
```

Example "Set Dispatch Script" Script

Below you find a sample "Set Dispatch Script" Script:

```
Set Field [gErrorCode, External("Serial-SetDispatchScript",
                       Status(CurrentFileName) & "| scriptkey=1")]
If [Left(gErrorCode, 2 ) = "$$"]
     Beep
     Show Message [An error occurred while setting the dispatch script]
     Halt Script
End If
```

Example Start Receiving Script

Below you find a sample "Start Receiving" script:

```
Perform Script [Sub-scripts, "Open Serial Port"]
Perform Script [Sub-scripts, "Set Dispatcher Script"]
```

When you want to begin receiving perform the "Start receiving script".

Serial-SetDispatchScript

Syntax Set Field[gResult, External("Serial-SetDispatchScript", "*filename* | scriptkey=*x* ") or
Set Field[gResult, External("Serial-SetDispatchScript", "")

Sets the Dispatch Script to call when data is received. If you give an empty parameter "", the Dispatch Script is removed.

Parameters

filename: the name of the file with the Dispatch Script
scriptkey=*x :* the key number in the menu of the Dispatch Script. *x* must be in the range from 0-9

Result

The returned result is an error code. An error always starts with 2 dollars, followed by the error code. You should always check for errors.

Returned error codes can be:

0	no error	the Dispatch Script was set
$$-50	paramErr	There was an error with the parameter

Other errors might be returned.

Example Usage

```
Set Field[ gErrorCode, External("Serial-SetDispatchScript",
                      Status(CurrentFileName) & "| scriptkey=1") ]
```

This will set the Dispatch Script to the script with shortcut control-1 (or command-1) of the current file.

Example Usage (resetting the Dispatch Script)

```
Set Field[ gErrorCode, External("Serial-SetDispatchScript", "") ]
```

This will reset the Dispatch Script. No action is taken when data is received.

Serial-DataWasReceived

Syntax Set Field[gResult, External("Serial-DataWasReceived", "")

Returns 1 when data was received on a serial port. Use this function to see if this is an event that needs to be handled.

Parameters

no parameters leave empty for future use.

Result

The returned result is an boolean value. Returned is either:

0 no data received
1 data was received in the buffer

When this function returns 1 you can get the data with the function **Serial-Receive**.

Example Usage

```
If[ External("Serial-DataWasReceived",  "") ]
     Perform Script [Sub-scripts, “Process Data Received”]
Else
     ... do something else
Endif
```

Serial-RestoreSituation

Syntax Set Field[gResult, External("Serial-RestoreSituation", "")]

Bring the database file that was in front, before the Dispatch Script was called, back to the front.

Parameters

no parameters leave empty for future use.

Result

The returned result is an error code:

0 no error

At the moment no other results are returned.

Example Usage

```
Set Field [gErrorCode, External("Serial-RestoreSituation", "") ]
```

Serial-ToASCII

Syntax Set Field[gResult, External("Serial-ToASCII", "asciiCode | asciiCode | asciiCode |...")]

Converts (one or more) numbers to their equivalent ASCII characters.

Parameters

ASCIIcode(s) one or more numbers in the range from 0-255.

Result

The returned result is the string of text of the ASCII codes.

Example Usage

```
Set Field [text, External("Serial-ToASCII", "65|65|80|13") ]
```

This will result in the text "**AAP<CR>**" where <CR> is a Carriage Return character.

NOTE You can also use hexadecimal notation for the numbers. Use 0x00 to 0xFF to indicate hexadecimal notation.

Example Usage

```
Set Field [text, External("Serial-ToASCII", "0x31|0x32|0x33|0x0D|0x0A") ]
```

This will result in the text "**123<CR><LF>**" where <CR> is a Carriage Return character and <LF> is a Line Feed character.

NOTE The graphic rendition of characters greater than 127 is undefined in the **A**merican **S**tandard **C**ode for **I**nformation **I**nterchange (ASCII Standard) and varies from font to font and from computer to computer and may look different when printed.